

# NAG Library Routine Document

## F11MLF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F11MLF computes the 1-norm, the  $\infty$ -norm or the maximum absolute value of the elements of a real, square, sparse matrix which is held in compressed column (Harwell–Boeing) format.

### 2 Specification

```
SUBROUTINE F11MLF (NORM, ANORM, N, ICOLZP, IROWIX, A, IFAIL)
INTEGER          N, ICOLZP(*), IROWIX(*), IFAIL
REAL (KIND=nag_wp) ANORM, A(*)
CHARACTER(1)     NORM
```

### 3 Description

F11MLF computes various quantities relating to norms of a real, sparse  $n$  by  $n$  matrix  $A$  presented in compressed column (Harwell–Boeing) format.

### 4 References

None.

### 5 Arguments

1: NORM – CHARACTER(1) *Input*

*On entry:* specifies the value to be returned in ANORM.

NORM = '1' or 'O'

The 1-norm  $\|A\|_1$  of the matrix is computed, that is  $\max_{1 \leq j \leq n} \sum_{i=1}^n |A_{ij}|$ .

NORM = 'I'

The  $\infty$ -norm  $\|A\|_\infty$  of the matrix is computed, that is  $\max_{1 \leq i \leq n} \sum_{j=1}^n |A_{ij}|$ .

NORM = 'M'

The value  $\max_{1 \leq i, j \leq n} |A_{ij}|$  (not a norm).

*Constraint:* NORM = '1', 'O', 'I' or 'M'.

2: ANORM – REAL (KIND=nag\_wp) *Output*

*On exit:* the computed quantity relating the matrix.

3: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

- 4: ICOLZP(\*) – INTEGER array *Input*  
**Note:** the dimension of the array ICOLZP must be at least  $N + 1$ .  
*On entry:* ICOLZP( $i$ ) contains the index in  $A$  of the start of a new column. See Section 2.1.3 in the F11 Chapter Introduction.
- 5: IROWIX(\*) – INTEGER array *Input*  
**Note:** the dimension of the array IROWIX must be at least  $ICOLZP(N + 1) - 1$ , the number of nonzeros of the sparse matrix  $A$ .  
*On entry:* the row index array of sparse matrix  $A$ .
- 6: A(\*) – REAL (KIND=nag\_wp) array *Input*  
**Note:** the dimension of the array  $A$  must be at least  $ICOLZP(N + 1) - 1$ , the number of nonzeros of the sparse matrix  $A$ .  
*On entry:* the array of nonzero values in the sparse matrix  $A$ .
- 7: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0,  $-1$  or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value  $-1$  or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N = \langle value \rangle$ .

Constraint:  $N \geq 0$ .

On entry,  $NORM = \langle value \rangle$ .

Constraint:  $NORM = '1', 'O', 'I'$  or  $'M'$ .

IFAIL =  $-99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL =  $-399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL =  $-999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

Not applicable.

## 8 Parallelism and Performance

F11MLF is not threaded in any implementation.

## 9 Further Comments

None.

## 10 Example

This example computes norms and maximum absolute value of the matrix  $A$ , where

$$A = \begin{pmatrix} 2.00 & 1.00 & 0 & 0 & 0 \\ 0 & 0 & 1.00 & -1.00 & 0 \\ 4.00 & 0 & 1.00 & 0 & 1.00 \\ 0 & 0 & 0 & 1.00 & 2.00 \\ 0 & -2.00 & 0 & 0 & 3.00 \end{pmatrix}.$$

### 10.1 Program Text

```

Program f11mlfe

!      F11MLF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: f11mlf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: anorm
Integer                    :: i, ifail, n, nnz
Character (1)              :: norm
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:)
Integer, Allocatable       :: icolzp(:), irowix(:)
!      .. Executable Statements ..
Write (nout,*) 'F11MLF Example Program Results'
!      Skip heading in data file
Read (nin,*)

!      Read order of matrix and number of right hand sides

Read (nin,*) n

Allocate (icolzp(n+1))

!      Read the matrix A

Read (nin,*) icolzp(1:n+1)
nnz = icolzp(n+1) - 1

Allocate (a(nnz),irowix(nnz))

Do i = 1, nnz
  Read (nin,*) a(i), irowix(i)
End Do

!      Calculate 1-norm

```

```

norm = '1'

!   ifail: behaviour on error exit
!           =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call f11mlf(norm,anorm,n,icolzp,irowix,a,ifail)

!   Output norm

Write (nout,*)
Write (nout,*) 'One-norm'
Write (nout,'(F7.3)') anorm

!   Calculate M-norm

norm = 'M'

ifail = 0
Call f11mlf(norm,anorm,n,icolzp,irowix,a,ifail)

!   Output norm

Write (nout,*)
Write (nout,*) 'Max'
Write (nout,'(F7.3)') anorm

!   Calculate I-norm

norm = 'I'

ifail = 0
Call f11mlf(norm,anorm,n,icolzp,irowix,a,ifail)

!   Output norm

Write (nout,*)
Write (nout,*) 'Infinity-norm'
Write (nout,'(F7.3)') anorm

End Program f11mlfe

```

## 10.2 Program Data

F11MLF Example Program Data

```

5  N
1
3
5
7
9
12  ICOLZP(I) I=1,..,N+1
2.  1
4.  3
1.  1
-2. 5
1.  2
1.  3
-1. 2
1.  4
1.  3
2.  4
3.  5  A(I), IROWIX(I) I=1,NNZ

```

### **10.3 Program Results**

F11MLF Example Program Results

One-norm  
6.000

Max  
4.000

Infinity-norm  
6.000

---